

Code Style

There Must Be No Labels or Goto Statements

- **Symbolic Name in XML Configuration:** checkNoLabelsOrGotoStatements
- **Dependant Tags in XML Configuration:** -

The check makes sure that there are no labels and goto statements in the test code.

There Must Be No Nested Alt Statements

- **Symbolic Name in XML Configuration:** checkNoNestedAltStatements
- **Dependant Tags in XML Configuration:** maximumAllowedNestingDepth

The check makes sure that there are no alt statements nested within other alt statements or altstep definitions beyond a given depth (specified via the **maximumAllowedNestingDepth** configuration tag). The topmost alt or the enclosing altstep definition is considered of depth 0, first level nesting is of depth 1, and so on. The default nesting depth is 0, meaning that no alt statements are allowed within other alt statements or altstep definitions.

Example:

```
module checNoNestedAltStatements {
  altstep as_0 ()
  runs on myComponent {
    //nesting within an altstep
    [] p2.receive ( msg_b ) {
    }
    [] p2.receive ( msg_c ) {
      setverdict ( pass );
      // nested alt, level 1
      alt {
        [] p2.receive ( msg_b ) {
        }
        [] p2.receive ( msg_c ) {
          setverdict ( pass );
        }
        [] p2.receive {
        }
      }
    }
  }
  [] p2.receive {
    // nested alt, level 1
    alt {
      [] p2.receive ( msg_b ) {
```


The check makes sure that there are no modified templates are derived from modified templates. In the example, myTemplate3 is a modified template of degree 2 and myTemplate4 is a modified template of degree 4. Both will be reported by T3Q.

Example:

```
module checkNoModifiedTemplateOfModifiedTemplate {
  type record MyRecordType {
    integer field1 optional,
    charstring field2,
    boolean field3
  }
  template MyRecordType MyTemplate1 := {
    field1 := 123,
    field2 := "A string",
    field3 := true
  }
  template MyRecordType MyTemplate2 modifies MyTemplate1 := {
    field1 := omit,
    field2 := "A modified string"
  }
  // MyTemplate2 is already a modified template
  template MyRecordType MyTemplate3 modifies MyTemplate2 := {
    field1 := 22
  }
  // this one is even modified two times
  template MyRecordType MyTemplate4 modifies MyTemplate3 := {
    field3 := false
  }
  // NESTED TEMPLATES
  type record MyRecordType2 {
    integer field1 optional,
    charstring field2,
    boolean field3,
    MyRecordType nestedTemplate
  }
  template MyRecordType MyNestedTemplate1 := {
    field1 := 123,
    field2 := "A string",
    field3 := true,
    nestedTemplate :={field1:= 1, field2:="a", field3:=true}
  }
  template MyRecordType MyNestedTemplate2 modifies MyNestedTemplate1 := {
    field2 := "B string",
    nestedTemplate :={field1:= 2, field2:="b", field3:=true}
  }
  template MyRecordType MyNestedTemplate3 modifies MyNestedTemplate2 := {
```

```

    field2 := "C string",
    nestedTemplate :={field1:= 3, field2:="c", field3:=true}
}
}

```

Local Definitions Must Be Declared at the Beginning of Testcases, Functions, Altsteps, and Component Definitions, in a Specified Order

- **Symbolic Name in XML Configuration:** checkLocalDefinitionsComeFirst
- **Dependant Tags in XML Configuration:** localDefinitionTypes

This is an evolved version of the check for variables only in the same context (declared before any other statements). In this version, not only variables, but also local constants and timers can be considered (if configured). Additionally, the order of occurrence of these three classes of local definitions can be specified and checked for. With the help of the help of the **localDefinitionTypes** configuration tag, which contains a list **string** tags, the contents and the order of local definitions can be specified. The default configuration contains four possible types of local definitions: **VarInstance** (variables), **ConstDef** (local constants), **TimerInstance** (local timers), and **PortInstance** (port instances, only within components), configured in this order, meaning that constants are expected to be declared after all variables and before all timers (and ports, in the context of component definitions):

```

<localDefinitionTypes>
  <string>VarInstance</string>
  <string>ConstDef</string>
  <string>TimerInstance</string>
  <string>PortInstance</string>
</localDefinitionTypes>

```

This order can be changed by moving the configured entries up and down the list. If timers and ports are to be disregarded, for example, then the whole configuration tags shall be omitted, so that the configuration will be:

```

<localDefinitionTypes>
  <string>VarInstance</string>
  <string>ConstDef</string>
</localDefinitionTypes>

```

Only the local definition types provided in the default configuration can be used, and only if spelled correctly, meaning that case-sensitivity matters in this context ("constdef" hence will not be recognized).

Note that control part definitions are not analyzed. This will be subject to change, where control part definitions will be analyzed as well.

Import Statements Must Be Declared at the Beginning of Modules

- **Symbolic Name in XML Configuration:** `checkImportsComeFirst`
- **Dependant Tags in XML Configuration:** -

Similarly to "Variables Must Be Declared at the Beginning of Testcases, Functions and Altsteps", this check makes sure that all import statements are always at the beginning of any module.

There Must Be No Duplicated Identifiers On the Module Level

- **Symbolic Name in XML Configuration:** `checkNoDuplicatedModuleDefinitionIdentifiers`
- **Dependant Tags in XML Configuration:** -

This check will make sure there are no identical identifiers for definitions on the module level among the analyzed modules (regardless of type).

There Must Be No Unused Definitions On the Module Level

- **Symbolic Name in XML Configuration:** `checkZeroReferencedModuleDefinitions`
- **Dependant Tags in XML Configuration:** `zeroReferencedModuleDefinitionsExcludedRegExp`

This check will make sure there are no unused definitions on the module level among the analyzed modules, including references through imports. References within import statements however will be disregarded, meaning that if a definition is imported, but never used, it will be considered an unused definition as well. Group definitions are excluded from this check. Additionally, through the **`zeroReferencedModuleDefinitionsExcludedRegExp`** configuration tag, a regular expression can be specified to allow certain modules (based on the module name) to be excluded from this check. The regular expression is empty by default, meaning that all modules are considered by default.

There Must Be No Inline Templates

- **Symbolic Name in XML Configuration:** `checkNoInlineTemplates`
- **Dependant Tags in XML Configuration:** -

This check will make sure there are no inline templates used in the analyzed modules.

There Must Be No Over-specific Runs On Clauses

- **Symbolic Name in XML Configuration:** `checkNoOverSpecificRunsOnClauses`

- **Dependant Tags in XML Configuration:** recursionInCheckNoOverSpecificRunsOnClauses, aliasInCheckNoOverSpecificRunsOnClauses (since v2.0.0b27)

This check will make sure no over-specific **runs on** clauses are used. An over-specific runs on clause is when none of the component element definitions (variables, timers, constants, and ports) of the component used in the runs on clause are referenced within the body of the function or altstep with the runs on clause. If at least one of the component element definitions is referenced within the body of the function or altstep, then no problem is reported. If the **recursionInCheckNoOverSpecificRunsOnClauses** setting is enabled (which it is by default), also the functions and altsteps referenced within the body of the current construct will be inspected. This is due to the fact that often wrapper functions are used that do not make direct use of the component element definitions themselves.

As of v1.0.3, test cases are not considered in this check, due to a change request based on the notion that test cases are required to have a runs on clause, regardless of whether they actually utilize any of the component element definitions of the MTC.

Example:

```

module checkNoOverSpecificRunsOn {
  //global const for validation
  const someType someGlobalConst := 21;
  //component in question
  type component someComponent {
    var someType someVarName := 2;
    var someType someOtherVarName := 2, someAnotherVarName := 4;
    var charstring someCharStringVar := "xyz";
    var template someType someTemplateVarName := 3;
    const someType someConstName := 1;
    const integer someIntegerConstName := 42;
    const someOtherType someConstName2 := someModuleParameterName2, someConstName3 := some
    timer someTimer;
    port somePortType somePortInstance;
  }
  type component someComponent2 extends someComponent{
    port somePortType somePortInstance2;
  }
  //good
  function someFunction ()
  runs on someComponent {
    //some references to component definitions here
    someTimer.start ( 10.0 );
  }
  //bad
  function someOverSpecificFunction ()
  runs on someComponent {
    //ay(); //this will make it valid
  }
}

```

```

        //wrapperfunction(); //this will make it valid
        wrapperFunction(); //unresolvable due to a typo - proper error message provide
//no references to component definitions here
}
    //irrelevant
function someGenericFunction () runs on someComponent{
    somePortInstance.send("x");
//no relation
}
//good
//a tricky example with a referenced function that uses the relevant fields
//wraper function
function wrapperfunction ()
runs on someComponent {
    //function that uses the component definitions
    someFunction ();
}
    //bad
testcase tcx ()
runs on someComponent {
    //ay(); //this will make it valid
}
    //good
    //tricky wrapper example
altstep ax ()
runs on someComponent {
    var integer a := someFunction();
    [] ay ()
    [] ay ()
}
    //good
altstep ay ()
runs on someComponent {
    var integer a := someFunction();
    [] someTimer.timeout {
        wrapperfunction();
    }
}
    //tricky cyclic call sequences
function f1() runs on someComponent{
    f2();
    f3()
}
function f2() runs on someComponent{
    f1()
    f4()
}
function f3() runs on someComponent{

```

```

        f2()
    }
    function f4() runs on someComponent{
    }
}

```

As of v2.0.0b27 an additional setting **aliasInCheckNoOverSpecificRunsOnClauses** is provided. It enables the special treatment of component "alias" type definitions (component type definitions without any owned definitions that extend another component type and inherit its definitions). If this setting is enabled (which it is by default), a function or an altstep specified to run on a given component type will only raise a warning if none of the definitions of the component type(s) that this component type extends directly (i.e. not considering the components extended by those components recursively) are used within the function or altstep. It supersedes the previously introduced **extendsInCheckNoOverSpecificRunsOnClauses** setting.

For example:

```

//base component
type component componentWithDefinition {
    timer definedTimer;
}

//extension with own definitions
type component directExtension extends componentWithDefinition {
    var integer directExtensionVariable;
}

//multi alias / extension ("pure", without own definitions)
type component indirectAlias extends directExtension {

}

//multi alias / extension ("pure", without own definitions)
type component pureAlias extends componentWithDefinition {

}

//always good - base component
function someFunctionOnBaseComponent ()
runs on componentWithDefinition {
    definedTimer.start ( 10.0 );
}

//bad - no definition from directExtension used
function someFunctionOnDirectAlias ()
runs on directExtension {
    definedTimer.start ( 10.0 );
}

```

```

}

//bad - no definition from aliased directExtension used
function someFunctionOnMultiAlias ()
runs on indirectAlias {
    definedTimer.start ( 10.0 );
}

//good - definition from aliased directExtension is used (directAliasVariable)
function someFunctionOnMultiAliasWithReferenceToVariable ()
runs on indirectAlias {
    definedTimer.start ( 10.0 );
    directExtensionVariable := 1;
}

//good - definition from aliased componentWithDefinition is used (definedTimer)
function someFunctionOnPureAlias ()
runs on pureAlias {
    definedTimer.start ( 10.0 );
}

```

There Must Be No Unused Imports

- **Symbolic Name in XML Configuration:** checkNoUnusedImports
- **Dependant Tags in XML Configuration:** -

This is a rather complex quality check, that seeks to identify unused imports. As a first step it checks whether the imported module exists. If it does not exist, it is either because the imported module is not a part of the analyzed input, or because the name of the imported module has been misspelled. Whatever the case, if the module cannot be resolved, a corresponding message is provided and the check for that particular import statement is finished. If the module can be resolved, depending on the context, the following situations are checked:

1. If an **all** keyword is used in a non-type-restrictive manner (i.e. not associated to a certain definition type), then it is checked whether there is at least one reference of any module definition from the imported module within the importing module. Whether there are exceptions specified in the import statement makes no difference in this case, because in such a scenario, it is no longer possible to have references pointing back to the imported definition and thus the results of this check remain correct.
2. If a non-specific type import is used (i.e. an **all** keyword is preceded by definition type keyword), then it is checked whether there is at least one reference of any module definition of the given type from the imported module within the importing module. Same conditions apply as in the non-type-restrictive use of the **all** keyword. If all groups are to be considered, then only the definitions within groups are considered,

those outside groups are not considered.

3. If a specific type import is used (i.e. a definition type keyword, followed by a (list of) identifiers), several checks are performed:
 1. Check if the identifier is resolvable. If it is not it means that either the definition is not present within the imported module or the identifier is perhaps misspelled.
 2. If the identifier is resolvable, it is checked whether the definition is actually within the imported module. It may be the case that the definition has already been imported from another module (see also [Documentation/T3Q/Quality-Checks/Code-Style](#)), in which case a corresponding message is provided and the actual references for that identifier are not checked any further. It may also be the case that the definition has already been imported from another module, but also does not exist in the module from which it is attempted to be imported again (the definition has been moved perhaps?!), in which case another corresponding message is provided, and no further analysis of the actual references is performed.
 3. Finally, if the identifier is uniquely and correctly resolvable, it is checked whether there are any references to that imported definition within the importing module. If a specific group is imported, it is checked whether any of the definitions within that group are referenced in the importing module instead.

Note that currently this check is rather computationally expensive and thus **disabled by default**. Once suitable optimizations are added, it will be re-enabled by default.

There Must Be No Unused Formal Parameters

- **Symbolic Name in XML Configuration:** `checkNoUnusedFormalParameters`
- **Dependant Tags in XML Configuration:** -

This check makes sure there are no unused formal parameters. There are several peculiarities associated with this quality check:

- External functions are excluded from this check since their formal parameters can never be used within their TTCN-3 definitions
- In the case of modified template definitions, all the formal parameters defined in the modified template (currently matched by name only) are not checked again in the modifying template. Only the additional formal parameters defined in the modifying template are checked in such a case.
- Type parametrization is currently left out, since type parametrization is moved to a package in TTCN-3 v4.1.1.
- In the case of cyclic call sequences, a corresponding *INFORMATION* message will be generated.

There Must Be No Unused Local Definitions

- **Symbolic Name in XML Configuration:** checkNoUnusedLocalDefinitions
- **Dependant Tags in XML Configuration:** -

This check makes sure there are no unused local definitions. This covers local constants, timers, variables, ports, template variables and local template definitions. Currently component definitions, function definitions, altstep definitions and testcase definitions are analyzed, as well as module control parts.

There Must Be No Uninitialised Local Variables

- **Symbolic Name in XML Configuration:** checkNoUninitialisedVariables
- **Dependant Tags in XML Configuration:** checkNoUninitialisedVariablesExclude

This check makes sure there are no uninitialised local variables. This covers local variables declared within statement blocks or module control parts. Variables declared within components are not considered as they may or may not be initialised depending on which other behaviour constructs have been executed. Instead, the data flow analysis is performed strictly within the scope of the top-level statement block. In the following example, the state of the various variables declared within a function is described for each statement using comments directly above the statement.

```
function f_ConditionalSpec() {
    var integer v_s0;
    var integer v_s1;
    var integer v_s2 := 1;
    var integer v_s3;
    var integer v_s4;
    var integer v_s5;

    //v_s2 is initialised upon declaration -> no warning
    if (v_s2 == 1) {
        //v_s3 is not initialised -> warning
        //v_s0 initialised within the conditional after this statement
        v_s0 := v_s3 + 1;

        //v_s1 is not initialised -> warning
        //v_s2 is initialised -> no warning
        //v_s3 initialised within the conditional after this statement
        v_s3 := v_s2 + 1 + v_s1;

        //v_s3 is initialised above -> no warning
        //v_s4 initialised within the conditional after this statement
        v_s4 := v_s3 + 1;
    } else {
```

```

        //v_s2 is initialised -> no warning
        //v_s4 initialised within the else branch after this statement
        v_s4 := v_s2 + 1;
    }

    //v_s3 is only initialised within one of the possible paths -> warning
    //v_s4 is initialised within both possible paths (within all branches of the condition)
    v_s5 := v_s3 + v_s4;
}

```

For variables initialised within branching behaviour, such as if-else if-else constructs, alt statements, and call statements, the data flow analysis evaluates whether each branch directly initialises the variables with absolute certainty. If the variable is initialised in only some of the branches, then a warning is raised as it cannot be ensured that a variable will be initialised before use during execution. This also applies to nested branches. For loops, it cannot be ensured that a variable will be initialised as a loop can be skipped if the loop condition evaluates to false. In such case, variables initialised within the loop are considered as such in the subsequent statements within the loop, however, for statements outside the loop, the variable is still considered to be not initialised (unless there is explicit initialisation outside the loop).

The **checkNoUninitialisedVariablesExclude** setting can be used to filter out warnings for variables of certain types. The meta-types enumerated, union, record of, record, set of, set, can be listed in order to exclude warnings for all variables of the corresponding meta-type from the output, e.g. exclude all variables of all record types. Additionally, user defined concrete types can be listed as well, so that only warnings for variables of a specific type, e.g. a specific record type, are excluded from the output. In the following example configuration, all warnings for all variables of set and set of types, as well as for all variables of the type MyRecordType are excluded from the output.

```

<checkNoUninitialisedVariablesExclude >
  <string>set</string>
  <string>set of</string>
  <string>MyRecordType</string>
</checkNoUninitialisedVariablesExclude >

```

The meta-types enumerated, union, record of, record, set of, set, are currently set to be excluded by default when a new configuration profile is generated.

There Must Be No Literals

- **Symbolic Name in XML Configuration:** checkNoLiterals
- **Dependant Tags in XML Configuration:** -

This check makes sure there are no literals used, except in module parameters, template definitions, and constant definitions. Note that while in both module-level constants and local constants literals are permitted, in the case of templates, only template definitions at the module level can contain literals. Note also that currently matching symbols, boolean values, verdicts, omit-values, enumerated values, and address-values are not considered *literals*. This may be a subject to change.

There Must Be No ValueOf Operations for Values

- **Symbolic Name in XML Configuration:** checkNoValueOfForValues
- **Dependant Tags in XML Configuration:** -

This check makes sure there are no **valueof** operations on "pure" values. This currently covers references to local variables, formal parameters, component variables, return types of functions, and defined templates.

```
function f_Work() {
    var float v_ToConvert := 2.0;
    var float v_ValueCms;
    var template float v_ToConvertTemplate := 2.0;
    //Being a value already the call to valueof is redundant -> warning
    v_ValueCms := f_ConvertInchesToCm(valueof(v_ToConvert));
    //Should be ok -> no warning
    v_ValueCms := f_ConvertInchesToCm(valueof(v_ToConvertTemplate));
}

function f_Work (
    float p_ToConvert,
    template float p_ToConvertTemplate
) return template float {
    var float v_ValueCms;
    //Being a value already the call to valueof is redundant -> warning
    v_ValueCms := f_ConvertInchesToCm(valueof(p_ToConvert));
    //Should be ok -> no warning
    v_ValueCms := f_ConvertInchesToCm(valueof(p_ToConvertTemplate));
}
```

There Must Be No AnyValueOrNone in List Values

- **Symbolic Name in XML Configuration:** checkNoAnyValueOrNoneInListValues
- **Dependant Tags in XML Configuration:** -

This check makes sure there are no **AnyValueOrNone** assignments to "list" values (assignments to values within a template or value of a record of or set of type). This currently covers references to local variables, formal parameters, component variables, return types of functions, and defined templates.

```

type record MyListElement_Type {
    integer x
    //...
}

type set MySetElement_Type {
    integer x
    //...
}

type record of MyListElement_Type MyList_Type;
type set of MySetElement_Type MySet_Type;

type component exampleComponent {
    var template (omit) MyListElement_Type cv_MyListElement2 := {};
    var template (present) MyList_Type cv_MyList;
}

function f_FunctionReturningElement1(
    //..
) return template MyListElement_Type {
    //...
}

function f_Examples(
    template MyListElement_Type p_MyListElement,
    template(present) MyListElement_Type prp_MyListElement,
    MyListElement_Type pv_MyListElement
) runs on exampleComponent {
    var MyListElement_Type v_MyListElement1 := {};
    var template (omit) MyListElement_Type v_MyListElement2 := {};
    var template (omit) MySetElement_Type v_MySetElement2 := {};
    var template (value) MyListElement_Type v_MyListElement3 := {};
    var template (present) MyListElement_Type v_MyListElement4 := {};
    var template MyListElement_Type v_MyListElement5 := {};
    var template (present) MyList_Type v_MyList;
    var template (value) MyList_Type v_MyListV;
    var template (present) MySet_Type v_MySet;
    var integer i := 1;
    v_MyList[i] := v_MyListElement1; // ok (SingleVarInstance)
    v_MyList[i] := v_MyListElement2; // shall cause an error/warning
    v_MyList[i] := cv_MyListElement2; // shall cause an error/warning
    cv_MyList[i] := cv_MyListElement2; // shall cause an error/warning
    cv_MyList[i] := v_MyListElement2; // shall cause an error/warning
    v_MySet[i] := v_MySetElement2; // shall cause an error/warning
    v_MyList[i] := v_MyListElement3; // ok (restriction=value)
    v_MyList[i] := v_MyListElement3.x // ok (restriction=value) + field
    v_MyList[i] := v_MyListElement4; // ok (restriction=present)
}

```

```

v_MyList[i] := v_MyListElement5; // shall cause an error/warning
v_MyList[i] := cr_MyListElement1; // shall cause an error/warning
v_MyList[i] := cr_MyListElement2; // ok (restriction=value)
v_MyList[i] := cr_MyListElement3(1) // ok (restriction=value) + param
v_MyList[i] := v_MyListV[1] // ok (restriction=value) + field
v_MyList[i] := valueof(cr_MyListElement3(1)) // ok (valueOf)

v_MyList[i] := f_FunctionReturningElement1(); // shall cause an error/warning
//extra: parameters
v_MyList[i] := p_MyListElement; // shall cause an error/warning
v_MyList[i] := pv_MyListElement; // ok (FormalValuePar)
v_MyList[i] := prp_MyListElement; // ok (restriction=present)
//extra: direct
v_MyList[i] := omit; // shall cause an error/warning
v_MyList[i] := *; // shall cause an error/warning
}

template (omit) MyListElement_Type cr_MyListElement1 := {};
template (value) MyListElement_Type cr_MyListElement2 := {};
template (value) MyListElement_Type cr_MyListElement3(integer p) := {};
const MyListElement_Type ccr_MyListElement2 := {};

template (present) MyList_Type cr_MyList (
  MyListElement_Type p_Param1, // ok (FormalValuePar)
  template (value) MyListElement_Type p_Param2, // ok (restriction=value)
  template (omit) MyListElement_Type p_Param3, // shall cause an error/warning
  template (present) MyListElement_Type p_Param4, // ok (restriction=present)
  template MyListElement_Type p_Param5 // shall cause an error/warning
) := {
  //...
  p_Param1,
  //...
  p_Param2,
  //...
  p_Param3, // shall cause an error
  //...
  p_Param4,
  //...
  {x:=p_Param2.x}, // ok, not directly assigned to li
  //...
  p_Param5, // shall cause an error
  //...
  cr_MyListElement1, // shall cause warning
  //...
  cr_MyListElement2, // ok (restriction=value)
  //...
  cr_MyListElement3(1), // ok (restriction=value) + param
  //...

```

```

    ccr_MyListElement2                                // ok (constant)
};

function f_Extras() {
    var template MyListElement_Type v_Element := *;
    var template MyListElement_Type v_Element_e := {};
    var template (present) MyListElement_Type v_Element_p := {};
    var template MyList_Type v_List1 := { -, v_Element }; // shall cause an error/
    var template MyList_Type v_List2 := { -, -, omit}; // shall cause an error/
    var template MyList_Type v_List1 := { -, v_Element_e }; // shall cause an error/
    var template MyList_Type v_List1 := { -, v_Element_p }; // shall be fine?
    // BUT
    var template MyList_Type v_List3 := { cr_MyListElement2, * }; // is allowed i.e. shall
    var template MyList_Type v_List3p := { cr_MyListElement3(1), * }; // is allowed i.e. s
    var template MyList_Type v_List4 := { * }; // is allowed too
    v_List4[0] := * //shall still cause an error/warning
    var template MyListElement_Type v_Element_r := v_Element_e; // should be ignored?
    var template MyList_Type v_List1 := { -, v_Element_r }; // shall cause an error/
    // this is not checked as it is indirect
    var template (present) MyListElement_Type v_Element_rp := v_Element_e; //should be i
    var template MyList_Type v_List1 := { -, v_Element_rp }; // shall cause an error
    // -> not possible dire
    // -> in case earlier i

    //extra
    var template (present) MyList_Type v_MyList;
    var template (value) MyList_Type v_MyListV;
    v_MyList := {}; // ok
    v_MyList := {v_MyListV[1]}; // ok
}

```

There Must Be No Uninitialized Fields in Templates

- **Symbolic Name in XML Configuration:** checkNoUninitializedFieldsInTemplates
- **Dependant Tags in XML Configuration:** checkNoUninitializedFieldsInTemplatesRecursion

This check makes sure there are no uninitialized fields in templates. This currently covers declared templates (on the module level only) for structured types declared in TTCN-3 and ASN.1 (see ASN.1 Support for details).

```

//manually defined types
type record EventDefinitions {
    EventIds eventIds,
    integer fieldint,
    boolean fieldbool,
    integer fieldextmandat,
    integer fieldextopt optional,

```

```

        boolean fieldextboolopt optional
    }
    //this is probably a union?
    type record EventId {
        A1 eventA1
    }
    type record A1 {
        boolean fieldeventA1
    }
    type record of EventId EventIds

template ( value ) EventDefinitions cs_EventDefinitions_eventA1_Errors (
    boolean p_FieldeventA1 := true ,
    integer p_Fieldint := 0 ,
    boolean p_Fieldbool := true , //integer p_Fieldextmandat := 1,
    integer p_Fieldextopt := 2
    /*,
    template (omit) boolean p_Fieldextboolopt := omit
    */
) := {
    eventIds := {
        eventA1 := {
            fieldeventA1 := p_FieldeventA1
        }

    },
    fieldint := p_Fieldint ,
    fieldbool := p_Fieldbool ,
    //fielextmandat := p_Fieldextmandat, -> warning for missing field
    fieldextopt := p_Fieldextopt
    //fieldextboolopt := p_Fieldextboolopt -> warning for missing field
}

//correct version: should not raise warinings
template ( value ) EventDefinitions cs_EventDefinitions_eventA1 (
    boolean p_FieldeventA1 := true ,
    integer p_Fieldint := 0 ,
    boolean p_Fieldbool := true,
    integer p_Fieldextmandat := 1 ,
    integer p_Fieldextopt := 2 ,
    template ( omit ) boolean p_Fieldextboolopt := omit
) := {
    eventIds := {
        eventA1 := {
            fieldeventA1 := p_FieldeventA1
        }

    },

```

```

        fieldint := p_Fieldint ,
        fieldbool := p_Fieldbool ,
        fieldextmandat := p_Fieldextmandat ,
        fieldextopt := p_Fieldextopt ,
        fieldextboolopt := p_Fieldextboolopt
    }

```

By default only the first level of modified templates is checked in case a template modifies another template or is assigned to another template, and the field specifications in the base template are considered. With the option **checkNoUnitializedFieldsInTemplatesRecursion** multiple levels of modified or assigned templates are considered.

```

module checkNoUnitializedFieldsInTemplatesRecursion {
    type record R {
        integer f1,
        integer f2,
        integer f3
    }

    template R A := {f1 := 1} //base template, not fully-specified -> warning

    template R B1 modifies A := {f2 := 2, f3:= 3} //fully-specified, inherits f1 -> no warning

    template R B2 modifies A := {f2 := 2} //not fully-specified, inherits f1, misses f3 -> warning

    template R C1 modifies B1 := {f1 := 3} //fully-specified, inherits f2, f3 -> no warning

    template R C2 modifies B2 := {f3 := 3} //fully-specified, inherits f1 from A, f2 from B2
    //-> no warning if checkNoUnitializedFieldsInTemplatesRecursion is true
    //-> warning if checkNoUnitializedFieldsInTemplatesRecursion is false (only f2 from B2 considered)

    template tr t_tr_mod_base_full_assigned (charstring p_c) := t_tr_mod_base_full("a", {}) //no warning

    template tr t_tr_mod_base_full (charstring p_c) modifies t_tr_base_full := {
        common := (p_c) //no warning
    }

    template tr t_tr_base_full (integer p_ind) := {
        common := "yes", //no warning
        indication := p_ind,
        version := 1
    }

    template tr t_tr_ref_ref_mod_c(charstring p_c) modifies t_tr_ref_ref := {
        common := (p_c) //no warning
    }
}

```

```

template tr t_tr_ref_mod_c(charstring p_c) modifies t_tr_ref := {
    common := (p_c) //no warning
}

template tr t_tr_ref_ref_mod(charstring p_c) modifies t_tr_ref_ref := {
    version := (p_c) //warning
}

template tr t_tr_ref_mod(charstring p_c) modifies t_tr_ref := {
    version := (p_c) //warning
}

template tr t_tr_ref_ref (integer p_i) := t_tr_ref(p_i) //warning

template tr t_tr_ref(integer p_i) := t_tr_base(p_i) //warning
template tr t_tr_ref_no_par(integer p_i) := t_tr_base //warning

template tr t_tr_base(integer p_ind) := {
//    common := "yes", //common not specified

    indication := p_ind,
    version := 1
}

type record tr {
    charstring common,
    integer indication,
    integer version
}

template rx_sub rx_sub_par(template rx_sub p_rx) := p_rx //skipped

template rx_sub t_rx := {
    version := 1,
    extensions := "x1",
    size := 3
}

type rx rx_sub
type record rx {
    integer version,
    charstring extensions,
    integer size
}
}

```

There Must Be No Zero or Multiple Fields in Union Templates

- **Symbolic Name in XML Configuration:** checkNoZeroOrMultipleFieldsInUnionTemplates
- **Dependant Tags in XML Configuration:** checkNoUninitializedFieldsInTemplatesRecursion

This check makes sure there are no zero or multiple initialized fields in union templates. This currently covers declared templates (on the module level only) for union types declared in TTCN-3 and in ASN.1 (see ASN.1 Support for details).

```
module unions {
  template tu tu_func_mod modifies tu_func := {
    //skipped
  }

  template tu tu_func_ref := tu_func()
    //skipped

  template tu tu_func_ref_np := tu_func

  template tu tu_func := f_tu()
  //no warning (no way to determine) -> skipped

  template tu tu_func := tu_base()

  template tu tu_base_mod_empty modifies tu_base := {
    //no warning
  }

  template tu tu_base_mod_c modifies tu_base := {
    common := "b" //no warning, same
  }

  template tu tu_base_mod modifies tu_base := {
    version := 0 //warning more than one / different
  }

  template tu tu_base := {
    common := "a"
  }

  function f_tu() return tu {
    return tu_base
  }

  type union tu {
    charstring common,
```

```
        integer indication,  
        integer version  
    }  
}
```

By default only the first level of modified templates is checked in case a template modifies another template or is assigned to another template, and the field specifications in the base template are considered. With the option **checkNoUnitializedFieldsInTemplatesRecursion** multiple levels of modified or assigned templates are considered. While the TTCN-3 semantics state that in the case of modified templates the newly assigned field replaces the field specified in the base template, the warning from this check can still be beneficial in highlighting cases where that may be unintentional.